Implicit Differentiation

Douglas Meade, Ronda Sanders, and Xian Wu
Department of Mathematics

Overview

This lab provides experience working with functions defined implicitly.

Maple Essentials

- The new Maple commands introduced in this lab are:

Command	Description	Example
`display`	combine one or more plots in a single plot; part of the `plots` package	display([P1,P2], title="My Graph");
`implicitdiff`	compute derivatives for implicitly-defined functions	Finding $\frac{dy}{dx}$: `implicitdiff(eq, y, x);` Finding $\frac{d^n y}{dx^n}$: `implicitdiff(eq, y, x$n);`
`implicitplot`	create graph of function defined implicitly; part of the `plots` package	`implicitplot(eq, x=a..b, y=c..d);`
`pointplot`	plots a single point; part of the `plots` package	`pointplot([a,b], symbolsize=15);`
`fsolve`	compute a solution of equations numerically	`fsolve({eq1,eq2}, {x,y});`
`with`	loads the contents of a Maple package	`with(plots):`
`eval`	evaluates a given expression at a given point	`eval(eq, {x=a, y=b});`

- The *ImplicitDifferentiation* maplet is available from the course website:

  `http://www.math.sc.edu/calclab/141L-F10/labs/` → ImplicitDifferentiation

Preparation

Review §3.5 Implicit Differentiation in Stewart and §4.4 in *Calclabs with Maple*.

Activities

1. The curve with equation $2(x^2 + y^2)^2 = 25(x^2 - y^2)$ is called a lemniscate.
   - Find the equation of the tangent line to this curve at the point $(3, 1)$. Then graph the curve, the point, and the tangent line together on one plot with a viewing window of $[-5, 5] \times [-4, 4]$. (Ex. 29 on page 213)
   - Find all points on the lemniscate where the tangent line is horizontal or vertical. (Ex. 39 on page 214)
2. Find $\frac{d^2y}{dx^2}$ and $\frac{d^3y}{dx^3}$ if y is defined implicitly by $y + \sin y = x$.

Example Problem

We will solve Example 2 on page 209 of Stewart together using Maple:

- Use implicit differentiation to find $\frac{dy}{dx}$ for the Folium of Descartes $x^3 + y^3 = 6xy$.
- Find an equation of the tangent line to the Folium of Descartes at the point $(3, 3)$. (Then graph the curve, the point, and the tangent line with a viewing window of $[-5, 5] \times [-5, 5]$ as shown in Figure 4 on page 210.)
- At what point(s) in the first quadrant is the tangent line to the Folium of Descartes horizontal? (At what points is the tangent line vertical?)

Steps:

1. First, load the Maple `plots` package. Without the contents of this package, much of what we do today will not work.
   > `with(plots):`
2. Assign our equation using ':='.
   > `eq:= x^3 + y^3 = 6*x*y;`
3. Find (and assign) the derivative using implicit differentiation. Since we want $\frac{dy}{dx}$, we input `y` and then `x`.
   > `dydx:= implicitdiff(eq, y, x);`
4. Find (and assign) the slope of the tangent line at the point (3, 3).
   > `m:= eval(dydx, {x=3, y=3});`
5. Find (and assign) the equation of the tangent line. Remember: $y = m(x - x_1) + y_1$.
   > `L:= m*(x - 3) + 3;`
6. Next, write (and assign) commands to plot the curve, the point, and the tangent line. Write the commands separately using ':' so Maple does not display the output yet. (In the first plot command, the option `numpoints=10000` will insure a smooth curve.)
   > `P1:= implicitplot(eq, x=-5..5, y=-5..5, numpoints=10000):`
   > `P2:= pointplot([3,3], color=green, symbolsize=15):`
   > `P3:= plot(L, x=-5..5, y=-5..5, color=blue, linestyle=dash):`
7. Use the `display` command to display the curve, point, and tangent line on a single plot.
   > `display([P1, P2, P3], title=''Figure 1'');`
8. From the graph, we can see that the tangent line would be horizontal at a point located approximately at $(2.5, 3.1)$. To find the point exactly, we need to find a point on the curve where $\frac{dy}{dx} = 0$. We can find this point using `fsolve`.
   > `fsolve({eq, dydx=0}, {x=2..3, y=3..4});`
9. From the graph, we can see that the tangent line would be vertical at a point located approximately at $(3.1, 2.5)$. To find the point exactly, we need to find a point on the curve where $\frac{dy}{dx}$ is undefined. That is, a point where the denominator of $\frac{dy}{dx}$ is 0 We can find this point using `fsolve`.
   > `fsolve({eq, denom(dydx)=0}, {x=3..4, y=2..3});`

Additional Notes

The ImplicitDifferentiation maplet provides additional practice finding the slope of a curve at a point.

Assignment

Exercises 31 and 38 on pages 213-214 of Stewart. For problem 38, you will need to zoom to get a good look at the horizontal tangents. For example, `y=1.6..1.7` is a good view to distinguish the necessary ranges for the top of the wagon. Exercise 12 on page 66 of *CalcLabs*.